## 1. The Subterranean Brachistochrone

A *Brachistochrone* is a frictionless track that connects two locations and along which an object can get from the first point to the second in minimum time under only the action of gravity. Above ground short-range Brachistochrones are well understood. Here we consider a subterranean version in which the start and end points are widely separated on the surface of the Earth. This version of the problem was first considered by several authors back in 1966 (see Cooper (1966a); Kirmser (1966); Venezian (1966); Mallett (1966); Laslett (1966); Cooper (1966b)). A very nice modern treatment can be found in Calkin (1999).

For any path, the time $T$ to traverse from one end to the other can be written quite simply as

$$T = \int \frac{ds}{v}$$

where $ds^2 = dx^2 + dy^2 = dr^2 + r^2 d\theta^2$ denotes the incremental squared arc length along the path and $v$ denotes the speed at each point along the path.

Instantaneous speed is determined from conservation of energy as follows.

The force due to gravity above the surface of the Earth is $F = -GMm/r^2$.

Below the surface, the effective mass of the Earth is reduced from $M$ to $M(r/R)^3$, where $R$ denotes the radius of the Earth and $r$ denotes the distance the track is from the center of the Earth.

Hence, the gravitational force below the Earth's surface is $F = -GMmr/R^3$.

The gravitational force is the negative of the gradient of the potential energy field.

Hence, the kinetic plus potential energy at radius $r$ and velocity $v$ is given by

$$\text{KE} + \text{PE} = \frac{1}{2}mv^2 + \frac{1}{2}\frac{GMm}{R^3}r^2.$$

At the start, $r = R$ and $v = 0$. Hence, by conservation of energy,

$$\frac{1}{2}mv^2 + \frac{1}{2}\frac{GMm}{R^3}r^2 = \frac{1}{2}\frac{GMm}{R}.$$

Solving for $v$, we get

$$v = \sqrt{\frac{GM}{R}\left(1 - \frac{r^2}{R^2}\right)}.$$

Hence, our integral for the time to traverse the path is given by

$$T = \int \frac{\sqrt{dr^2 + r^2 d\theta^2}}{\sqrt{\frac{GM}{R}\left(1 - \frac{r^2}{R^2}\right)}} = \sqrt{\frac{R^3}{GM}} \int \frac{\sqrt{r'^2 + r^2}}{\sqrt{R^2 - r^2}} d\theta.$$

Following standard notational conventions from the calculus of variations, let $L(r, r')$ denote the integrand in the right-hand integral above. Because $L$ depends only on $r$ and $r'$ and not on $\theta$, we can use the Beltrami equation to describe a minimizer of this integral:

$$L - r' \frac{\partial L}{\partial r'} = C$$

where $C$ is an arbitrary constant. For our particular problem, we compute explicitly as follows:

$$\begin{aligned} L - r' \frac{\partial L}{\partial r'} &= \frac{\sqrt{r'^2 + r^2}}{\sqrt{R^2 - r^2}} - r' \frac{r'}{\sqrt{r'^2 + r^2}\sqrt{R^2 - r^2}} \\ &= \frac{r^2}{\sqrt{r'^2 + r^2}\sqrt{R^2 - r^2}} = C. \end{aligned}$$

Solving for $r'^2$, we get

$$r'^2 = r^2 \left( \frac{r^2 - C^2(R^2 - r^2)}{C^2(R^2 - r^2)} \right).$$

Introducing a new notation for the constant,

$$r_0 = \frac{C}{\sqrt{1 + C^2}} R,$$

we can simplify the formula for the square of the derivative:

$$\left(\frac{dr}{d\theta}\right)^2 = r^2 \, \frac{R^2}{r_0^2} \, \frac{r^2 - r_0^2}{R^2 - r^2}.$$

Taking the (positive) square root of both sides and isolating the $\theta$ dependant quantities from the $r$ dependant ones, we get

$$\frac{R}{r_0} \, d\theta = \sqrt{\frac{R^2 - r^2}{r^2 - r_0^2}} \, \frac{dr}{r}.$$

So, to arrive at our formula relating $r$ to $\theta$, we integrate:

$$\frac{R}{r_0}\,\theta = \int \sqrt{\frac{R^2 - r^2}{r^2 - r_0^2}}\;\frac{dr}{r}.$$

All that remains is to compute this integral explicitly. To this end, we make the following change of variable:

$$u = \sqrt{\frac{r^2 - r_0^2}{R^2 - r^2}}.$$

To compute $dr$, it is helpful to solve for $r$ as a function of $u$:

$$r = \sqrt{\frac{u^2 R^2 + r_0^2}{1 + u^2}}.$$

For $dr$, we get

$$dr = \frac{u}{r}\;\frac{R^2 - r_0^2}{(1 + u^2)^2}\;du.$$

Putting this altogether, we compute the integral as follows:

$$\begin{aligned}
\int \sqrt{\frac{R^2 - r^2}{r^2 - r_0^2}}\;\frac{dr}{r} &= \int \frac{1}{u}\;\frac{u}{r^2}\;\frac{R^2 - r_0^2}{(1 + u^2)^2}\;du \\
&= \int \frac{1 + u^2}{u^2 R^2 + r_0^2}\;\;\frac{R^2 - r_0^2}{(1 + u^2)^2}\;du \\
&= \int \frac{R^2 - r_0^2}{(u^2 R^2 + r_0^2)(1 + u^2)}\;du \\
&= \int \frac{R^2 + u^2 R^2 - u^2 R^2 - r_0^2}{(u^2 R^2 + r_0^2)(1 + u^2)}\;du \\
&= R^2 \int \frac{du}{u^2 R^2 + r_0^2}\;du - \int \frac{du}{1 + u^2}\;du \\
&= \frac{R}{r_0}\;\tan^{-1}\frac{Ru}{r_0} - \tan^{-1} u \;+\; C \\
&= \frac{R}{r_0}\;\tan^{-1}\frac{R}{r_0}\sqrt{\frac{r^2 - r_0^2}{R^2 - r^2}} - \tan^{-1}\sqrt{\frac{r^2 - r_0^2}{R^2 - r^2}} \;+\; C
\end{aligned}$$

Hence, our equation relating $\theta$ to $r$ can now be written as

$$\theta = \tan^{-1}\frac{R}{r_0}\sqrt{\frac{r^2 - r_0^2}{R^2 - r^2}} - \frac{r_0}{R}\;\tan^{-1}\sqrt{\frac{r^2 - r_0^2}{R^2 - r^2}} \;+\; C'.$$

Without loss of generality, we may assume that $C' = 0$.

If we let $r = r_0$, then $\theta = 0$.

If we let $r = R$, then $\theta = \left(1 - \frac{r_0}{R}\right) \frac{\pi}{2}$.

The formula as derived has one endpoint at the surface of the Earth and the other endpoint at the nader given by $r = r_0$. The entire path starts at the surface and returns to the surface. In other words, it is two curves of the form shown. Hence, the angular extent of the path, $\Delta\theta$ is simply

$$\Delta\theta = \left(1 - \frac{r_0}{R}\right) \pi.$$

*Example.* For a tunnel that extends $45°$, we have $r_0 = \frac{3}{4}R$.

## 2. Solution Via Numerical Optimization

## 3. Comparison Between Numerical Computation and Exact Result

```
param pi := 4*atan(1);
param pi2 := pi/2;
param eps := 1e-15;

param n := 512;

param G := 6.67384e-11;  # m^3 / kg s^2
param M := 5.972e+24;    # kg
param R := 6.371e+6;     # m

param theta {j in 0..n};
param dtheta {j in 1..n} := (theta[j] - theta[j-1]);

var r    {j in 0..n} >= 0;
var rmid {j in 1..n} = (r[j]+r[j-1])/2;

var dr {j in 1..n} = (r[j] - r[j-1]);
var ds {j in 1..n} = sqrt( dr[j]^2 + (rmid[j]*dtheta[j])^2 );

var dt {j in 1..n} = 2*ds[j] / ( sqrt(eps+1-(r[j]/R)^2) + sqrt(eps+1-(r[j-1]/R)^2) );

minimize time: sqrt(R/(2*G*M)) * sum {j in 1..n} dt[j];

fix r[0] := R;
fix r[n] := R;
let {j in 0..n} r[j] := (j/n)*r[n] + (1-j/n)*r[0] - 2.0*R*(j/n)*(1-j/n);

let theta[0] := pi/2;
let theta[n] := pi/4;
#use a nonlinear interpolation that bunches near the endpoints
let {j in 0..n} theta[j] := sin(pi2*j/n)^2*theta[n] + cos(pi2*j/n)^2*theta[0];

option loqo_options "verbose=2 iterlim=2000 sigfig=12 inftol=1.3e-12";
solve;
```




Fig. 1.— Tunnel paths spanning $10$, $30$, $45$, $60$, $90$, and $120$ degrees. The numerically obtained solution is shown as a solid line. The dashed line of the same color is the exact solution. Note that the two solutions match with high percision in all cases except the $120$ degree example.

## REFERENCES

M.G. Calkin. *Lagrangian and Hamilonian Mechanics: Solution to the Exercises*. World Scientific, 1999.

P.W. Cooper. Through the Earth in Forty Minutes. *Am. J. Phys.*, 34:68–70, 1966a.

P.W. Cooper. Further Commentary on Through the Earth in Forty Minutes. *Am. J. Phys.*, 34: 703–704, 1966b.

P.G. Kirmser. An Example of the Need for Adequate References. *Am. J. Phys.*, 34:701, 1966.

L.J. Laslett. Trajectory for Minimum Transit Time Through the Earth. *Am. J. Phys.*, 34:702–703, 1966.

R.L. Mallett. Comments on Through the Earth in Forty Minutes. *Am. J. Phys.*, 34:702, 1966.

G. Venezian. Terrestrial Brachistochrone. *Am. J. Phys.*, 34:701, 1966.

---

This manuscript was prepared with the AAS LaTeX macros v5.2.